In[1]:=
```
SetDirectory["~/Factorisation/"];
<< kappaLib.m
<< helper.m
```

```
KappaLib v1.1
Loading helper.m..
```

■ Metaclass VI:

In[4]:=
```
vars = {x0, x1, x2, x3};
```

In[5]:=
```
kappa = emMatrixToKappa[{{a1, 0, 0, -b1, 0, 0}, {0, a2, 0, 0, a4, 0}, {0, 0, a3, 0, 0, a5}, {b1, 0, 0, a1, 0, 0}, {0, a4, 0, 0, a2, 0}, {0, 0, a5, 0, 0, a3}}];
```

In[6]:=
```
vars = {x0, x1, x2, x3};
fr = emKappaToFresnel[kappa, vars];
```

■ We may assume that a4, a5 != 0

In[8]:=
```
FullSimplify[fr /. {a4 → 0, x1 → 0, x3 → 0}]
FullSimplify[fr /. {a5 → 0, x0 → 0, x3 → 0}]
```

Out[8]=
```
0
```

Out[9]=
```
0
```

In[10]:=
```
frSym = x0^4 + x1^4 - x2^4 - x3^4 + D0 x0 x1 x2 x3 +
    D1 (x2^2 x3^2 - x0^2 x1^2) - D2 (x1^2 x3^2 + x0^2 x2^2) - D3 (x1^2 x2^2 + x0^2 x3^2);
```

In[11]:=
```
subSym = {
    D0 -> (2 a1^2 a2/(a4 a5 b1) - 2 a1 a2^2/(a4 a5 b1) - 2 a1^2 a3/(a4 a5 b1) + 2 a2^2 a3/(a4 a5 b1) + 2 a1 a3^2/(a4 a5 b1) -
       2 a2 a3^2/(a4 a5 b1) + 2 a1 a4/(a5 b1) - 2 a3 a4/(a5 b1) - 2 a1 a5/(a4 b1) + 2 a2 a5/(a4 b1) + 2 a2 b1/(a4 a5) - 2 a3 b1/(a4 a5)),
    D1 -> ((a2 - a3)^2 - a4^2 - a5^2)/(a4 a5),
    D2 -> ((a1 - a3)^2 - (a5^2 - b1^2))/(a5 b1),
    D3 -> ((a1 - a2)^2 - (a4^2 - b1^2))/(a4 b1)
  };
```

In[12]:=
```
Simplify[fr - (a4 a5 b1) frSym /. subSym]
```

Out[12]=
```
0
```

■ We assume that the Fresnel polynomial factorises

In[13]:=
```
A = Table[ToExpression["A" <> ToString[Min[{i, j}]] <> ToString[Max[{i, j}]]],
   {i, 0, 3}, {j, 0, 3}];
B = Table[ToExpression["B" <> ToString[Min[{i, j}]] <> ToString[Max[{i, j}]]],
   {i, 0, 3}, {j, 0, 3}];
A // MatrixForm
B // MatrixForm
factorised = (vars.A.vars) (vars.B.vars);
```

Out[15]//MatrixForm=

$$\begin{pmatrix} A00 & A01 & A02 & A03 \\ A01 & A11 & A12 & A13 \\ A02 & A12 & A22 & A23 \\ A03 & A13 & A23 & A33 \end{pmatrix}$$

Out[16]//MatrixForm=

$$\begin{pmatrix} B00 & B01 & B02 & B03 \\ B01 & B11 & B12 & B13 \\ B02 & B12 & B22 & B23 \\ B03 & B13 & B23 & B33 \end{pmatrix}$$

In[18]:=
```
cons = Union[Flatten[CoefficientList[frSym - factorised, vars]]];
```

In[19]:=
```
cons = simp[cons];
show[cons] // MatrixForm
```

Out[20]//MatrixForm=

1	0
2	1 − A00 B00
3	1 − A11 B11
4	−1 − A22 B22
5	−1 − A33 B33
6	−2 (A01 B00 + A00 B01)
7	−2 (A02 B00 + A00 B02)
8	−2 (A03 B00 + A00 B03)
9	−2 (A11 B01 + A01 B11)
10	−2 (A12 B11 + A11 B12)
11	−2 (A13 B11 + A11 B13)
12	−2 (A22 B02 + A02 B22)
13	−2 (A22 B12 + A12 B22)
14	−2 (A23 B22 + A22 B23)
15	−2 (A33 B03 + A03 B33)
16	−2 (A33 B13 + A13 B33)
17	−2 (A33 B23 + A23 B33)
18	−A11 B00 − 4 A01 B01 − A00 B11 − D1
19	−A33 B22 − 4 A23 B23 − A22 B33 + D1
20	−A22 B00 − 4 A02 B02 − A00 B22 − D2
21	−A33 B11 − 4 A13 B13 − A11 B33 − D2
22	−A22 B11 − 4 A12 B12 − A11 B22 − D3
23	−A33 B00 − 4 A03 B03 − A00 B33 − D3
24	−2 (A12 B00 + 2 A02 B01 + 2 A01 B02 + A00 B12)
25	−2 (2 A12 B01 + A11 B02 + A02 B11 + 2 A01 B12)
26	−2 (A13 B00 + 2 A03 B01 + 2 A01 B03 + A00 B13)
27	−2 (2 A13 B01 + A11 B03 + A03 B11 + 2 A01 B13)
28	−2 (A22 B01 + 2 A12 B02 + 2 A02 B12 + A01 B22)
29	−2 (A23 B00 + 2 A03 B02 + 2 A02 B03 + A00 B23)
30	−2 (2 A23 B02 + A22 B03 + A03 B22 + 2 A02 B23)
31	−2 (A23 B11 + 2 A13 B12 + 2 A12 B13 + A11 B23)
32	−2 (2 A23 B12 + A22 B13 + A13 B22 + 2 A12 B23)
33	−2 (A33 B01 + 2 A13 B03 + 2 A03 B13 + A01 B33)
34	−2 (A33 B02 + 2 A23 B03 + 2 A03 B23 + A02 B33)
35	−2 (A33 B12 + 2 A23 B13 + 2 A13 B23 + A12 B33)
36	−4 (A23 B01 + A13 B02 + A12 B03 + A03 B12 + A02 B13 + A01 B23) + D0

■ Equation (2): By renaming and scaling, we may assume that A00 = 1.

In[21]:=
```
sub = {A00 -> 1, B00 → 1};
```

In[22]:=
```
cons = simp[cons //. sub];
show[cons]
```

Out[23]//MatrixForm=

$$\begin{pmatrix}
1 & : & 0 \\
2 & : & 1 - A11\ B11 \\
3 & : & -1 - A22\ B22 \\
4 & : & -1 - A33\ B33 \\
5 & : & -2\ (A01 + B01) \\
6 & : & -2\ (A02 + B02) \\
7 & : & -2\ (A03 + B03) \\
8 & : & -2\ (A11\ B01 + A01\ B11) \\
9 & : & -2\ (A12\ B11 + A11\ B12) \\
10 & : & -2\ (A13\ B11 + A11\ B13) \\
11 & : & -2\ (A22\ B02 + A02\ B22) \\
12 & : & -2\ (A22\ B12 + A12\ B22) \\
13 & : & -2\ (A23\ B22 + A22\ B23) \\
14 & : & -2\ (A33\ B03 + A03\ B33) \\
15 & : & -2\ (A33\ B13 + A13\ B33) \\
16 & : & -2\ (A33\ B23 + A23\ B33) \\
17 & : & -A11 - 4\ A01\ B01 - B11 - D1 \\
18 & : & -A22 - 4\ A02\ B02 - B22 - D2 \\
19 & : & -A33 - 4\ A03\ B03 - B33 - D3 \\
20 & : & -A33\ B22 - 4\ A23\ B23 - A22\ B33 + D1 \\
21 & : & -A33\ B11 - 4\ A13\ B13 - A11\ B33 - D2 \\
22 & : & -A22\ B11 - 4\ A12\ B12 - A11\ B22 - D3 \\
23 & : & -2\ (A12 + 2\ A02\ B01 + 2\ A01\ B02 + B12) \\
24 & : & -2\ (A13 + 2\ A03\ B01 + 2\ A01\ B03 + B13) \\
25 & : & -2\ (A23 + 2\ A03\ B02 + 2\ A02\ B03 + B23) \\
26 & : & -2\ (2\ A12\ B01 + A11\ B02 + A02\ B11 + 2\ A01\ B12) \\
27 & : & -2\ (2\ A13\ B01 + A11\ B03 + A03\ B11 + 2\ A01\ B13) \\
28 & : & -2\ (A22\ B01 + 2\ A12\ B02 + 2\ A02\ B12 + A01\ B22) \\
29 & : & -2\ (2\ A23\ B02 + A22\ B03 + A03\ B22 + 2\ A02\ B23) \\
30 & : & -2\ (A23\ B11 + 2\ A13\ B12 + 2\ A12\ B13 + A11\ B23) \\
31 & : & -2\ (2\ A23\ B12 + A22\ B13 + A13\ B22 + 2\ A12\ B23) \\
32 & : & -2\ (A33\ B01 + 2\ A13\ B03 + 2\ A03\ B13 + A01\ B33) \\
33 & : & -2\ (A33\ B02 + 2\ A23\ B03 + 2\ A03\ B23 + A02\ B33) \\
34 & : & -2\ (A33\ B12 + 2\ A23\ B13 + 2\ A13\ B23 + A12\ B33) \\
35 & : & -4\ (A23\ B01 + A13\ B02 + A12\ B03 + A03\ B12 + A02\ B13 + A01\ B23) + D0
\end{pmatrix}$$

In[24]:=
```
tmp = Join[Take[cons, {5, 7}], Take[cons, {17, 19}], Take[cons, {23, 25}]];
tmp // MatrixForm
```

Out[25]//MatrixForm=

$$\begin{pmatrix}
-2\ (A01 + B01) \\
-2\ (A02 + B02) \\
-2\ (A03 + B03) \\
-A11 - 4\ A01\ B01 - B11 - D1 \\
-A22 - 4\ A02\ B02 - B22 - D2 \\
-A33 - 4\ A03\ B03 - B33 - D3 \\
-2\ (A12 + 2\ A02\ B01 + 2\ A01\ B02 + B12) \\
-2\ (A13 + 2\ A03\ B01 + 2\ A01\ B03 + B13) \\
-2\ (A23 + 2\ A03\ B02 + 2\ A02\ B03 + B23)
\end{pmatrix}$$

In[26]:=
```
Solve[toEqs[%], {B01, B02, B03, B11, B22, B33, B12, B13, B23}]
```

Out[26]=
$$\left\{\left\{B01 \to -A01,\ B02 \to -A02,\ B03 \to -A03,\ B11 \to 4\ A01^2 - A11 - D1,\ B22 \to 4\ A02^2 - A22 - D2,\right.\right.$$
$$\left.\left.B33 \to 4\ A03^2 - A33 - D3,\ B12 \to 4\ A01\ A02 - A12,\ B13 \to 4\ A01\ A03 - A13,\ B23 \to 4\ A02\ A03 - A23\right\}\right\}$$

In[27]:=
```
sub = Join[sub, %[[1]]]
```

Out[27]=
$$\left\{A00 \to 1,\ B00 \to 1,\ B01 \to -A01,\ B02 \to -A02,\ B03 \to -A03,\right.$$
$$B11 \to 4\ A01^2 - A11 - D1,\ B22 \to 4\ A02^2 - A22 - D2,\ B33 \to 4\ A03^2 - A33 - D3,$$
$$\left.B12 \to 4\ A01\ A02 - A12,\ B13 \to 4\ A01\ A03 - A13,\ B23 \to 4\ A02\ A03 - A23\right\}$$

In[28]:=
```
cons = simp[cons //. sub];
show[cons]
```

Out[29]//MatrixForm=
$$\begin{pmatrix}
1 & : & 0 \\
2 & : & 2\ A01\ \left(-4\ A01^2 + 2\ A11 + D1\right) \\
3 & : & 1 + A11\ \left(-4\ A01^2 + A11 + D1\right) \\
4 & : & 2\ A02\ \left(-4\ A02^2 + 2\ A22 + D2\right) \\
5 & : & 2\ A03\ \left(-4\ A03^2 + 2\ A33 + D3\right) \\
6 & : & -1 + A22\ \left(-4\ A02^2 + A22 + D2\right) \\
7 & : & -1 + A33\ \left(-4\ A03^2 + A33 + D3\right) \\
8 & : & 8\ A02\ A12 + 2\ A01\ \left(-12\ A02^2 + 2\ A22 + D2\right) \\
9 & : & 8\ A03\ A13 + 2\ A01\ \left(-12\ A03^2 + 2\ A33 + D3\right) \\
10 & : & 8\ A03\ A23 + 2\ A02\ \left(-12\ A03^2 + 2\ A33 + D3\right) \\
11 & : & -24\ A01^2\ A02 + 8\ A01\ A12 + 2\ A02\ (2\ A11 + D1) \\
12 & : & -24\ A01^2\ A03 + 8\ A01\ A13 + 2\ A03\ (2\ A11 + D1) \\
13 & : & -24\ A02^2\ A03 + 8\ A02\ A23 + 2\ A03\ (2\ A22 + D2) \\
14 & : & -8\ A01\ A02\ A11 - 8\ A01^2\ A12 + 2\ A12\ (2\ A11 + D1) \\
15 & : & -8\ A01\ A03\ A11 - 8\ A01^2\ A13 + 2\ A13\ (2\ A11 + D1) \\
16 & : & -8\ A02^2\ A12 - 8\ A01\ A02\ A22 + 2\ A12\ (2\ A22 + D2) \\
17 & : & -8\ A02\ A03\ A22 - 8\ A02^2\ A23 + 2\ A23\ (2\ A22 + D2) \\
18 & : & -8\ A03^2\ A13 - 8\ A01\ A03\ A33 + 2\ A13\ (2\ A33 + D3) \\
19 & : & -8\ A03^2\ A23 - 8\ A02\ A03\ A33 + 2\ A23\ (2\ A33 + D3) \\
20 & : & 8\ A03\ A12 + 8\ A02\ A13 + 8\ A01\ (-6\ A02\ A03 + A23) + D0 \\
21 & : & -4\ A02^2\ A11 - 16\ A01\ A02\ A12 + 4\ A12^2 - 4\ A01^2\ A22 + 2\ A11\ A22 + A22\ D1 + A11\ D2 - D3 \\
22 & : & -4\ A03^2\ A11 - 16\ A01\ A03\ A13 + 4\ A13^2 - 4\ A01^2\ A33 + 2\ A11\ A33 + A33\ D1 - D2 + A11\ D3 \\
23 & : & -4\ A03^2\ A22 - 16\ A02\ A03\ A23 + 4\ A23^2 - 4\ A02^2\ A33 + 2\ A22\ A33 + D1 + A33\ D2 + A22\ D3 \\
24 & : & 2\ \left(-4\ A02^2\ A13 - 4\ A01\ A03\ A22 + 2\ A13\ A22 + 4\ A12\ A23 - 8\ A02\ (A03\ A12 + A01\ A23) + A13\ D2\right) \\
25 & : & 2\ \left(-4\ A03^2\ A12 + 4\ A13\ A23 - 8\ A03\ (A02\ A13 + A01\ A23) - 4\ A01\ A02\ A33 + 2\ A12\ A33 + A12\ D3\right) \\
26 & : & 2\ \left(-8\ A01\ A03\ A12 + 4\ A12\ A13 - 4\ A02\ (A03\ A11 + 2\ A01\ A13) - 4\ A01^2\ A23 + 2\ A11\ A23 + A23\ D1\right.
\end{pmatrix}$$

■ Eliminate

In[30]:=
```
Variables[cons]
```

Out[30]=
$$\{A01, A02, A03, A11, A12, A13, A22, A23, A33, D0, D1, D2, D3\}$$

In[31]:=
```
elimVars = Variables[A]
condVars = {D0, D1, D2, D3}
```

Out[31]= {A00, A01, A02, A03, A11, A12, A13, A22, A23, A33}

Out[32]= {D0, D1, D2, D3}

In[33]:=
```
gb = FullSimplify[GroebnerBasis[cons, condVars, elimVars]]; // Timing
```

Out[33]= {10.39, Null}

In[34]:=
```
show[simp[gb]]
```

Out[34]//MatrixForm=

$$\begin{pmatrix} 1 & : & D0\left(4+D2^2\right)\left(4+D3^2\right) \\ 2 & : & D0\left(-4+D1^2\right)\left(4+D2^2\right) \\ 3 & : & D0\left(-4+D1^2\right)\left(4+D3^2\right) \\ 4 & : & \left(-4+D1^2\right)\left(4+D2^2\right)\left(4+D3^2\right) \\ 5 & : & \left(4+D2^2\right)(D1\ D2+2\ D3)\left(4+D3^2\right) \\ 6 & : & \left(4+D2^2\right)(2\ D2+D1\ D3)\left(4+D3^2\right) \\ 7 & : & \left(-4+D1^2\right)\left(4+D2^2\right)(2\ D1-D2\ D3) \\ 8 & : & \left(-4+D1^2\right)(2\ D1-D2\ D3)\left(4+D3^2\right) \\ 9 & : & \left(4+D2^2\right)(D2-D3)(D2+D3)\left(4+D3^2\right) \\ 10 & : & D0^2+4\left(-4+D1^2-D2^2-D1\ D2\ D3-D3^2\right) \end{pmatrix}$$

■ Equation (1) implies that D0 = 0. Equation (4) implies that D1 = +2 or -2.

In[35]:=
```
dSub = {D0 → 0, D1 → 2 sigma};
```

In[36]:=
```
gb = simp[gb /. dSub];
show[gb]
```

Out[37]//MatrixForm=

$$\begin{pmatrix} 1 & : & 0 \\ 2 & : & 4\left(4+D2^2\right)\left(4+D3^2\right)\left(-1+sigma^2\right) \\ 3 & : & 2\left(4+D2^2\right)\left(4+D3^2\right)(D3+D2\ sigma) \\ 4 & : & 2\left(4+D2^2\right)\left(4+D3^2\right)(D2+D3\ sigma) \\ 5 & : & \left(4+D2^2\right)(D2-D3)(D2+D3)\left(4+D3^2\right) \\ 6 & : & -4\left(4+D2^2\right)(D2\ D3-4\ sigma)\left(-1+sigma^2\right) \\ 7 & : & -4\left(4+D3^2\right)(D2\ D3-4\ sigma)\left(-1+sigma^2\right) \\ 8 & : & -4\left(4+D2^2+D3^2+2\ D2\ D3\ sigma-4\ sigma^2\right) \end{pmatrix}$$

■ Equation (3): (D3+D2 sigma)=0

In[38]:=
```
dSub = Append[dSub, D2 → -sigma D3];
```

In[39]:=
```
gb = simp[gb /. dSub];
show[gb]
```

Out[40]//MatrixForm=

$$\begin{pmatrix} 1 & : & 0 \\ 2 & : & 4\left(4+\text{D3}^2\right)\left(-1+\text{sigma}^2\right) \\ 3 & : & 4\left(4+\text{D3}^2\right)^2\,\text{sigma}\left(-1+\text{sigma}^2\right) \\ 4 & : & 4\left(4+\text{D3}^2\right)\left(-1+\text{sigma}^2\right)\left(4+\text{D3}^2\,\text{sigma}^2\right) \\ 5 & : & \text{D3}^2\left(4+\text{D3}^2\right)\left(-1+\text{sigma}^2\right)\left(4+\text{D3}^2\,\text{sigma}^2\right) \\ 6 & : & -2\,\text{D3}\left(4+\text{D3}^2\right)\left(-1+\text{sigma}^2\right)\left(4+\text{D3}^2\,\text{sigma}^2\right) \\ 7 & : & 4\left(4+\text{D3}^2\right)\text{sigma}\left(-1+\text{sigma}^2\right)\left(4+\text{D3}^2\,\text{sigma}^2\right) \end{pmatrix}$$

In[41]:=
```
dSub
```

Out[41]=
```
{D0 → 0, D1 → 2 sigma, D2 → -D3 sigma}
```

- **We have proven that:**

  D0 = 0, D1 = sigma x 2, D2 = -sigma D3

  where sigma = +1 or sigma = -1

- **Metrics for general sigma**

In[42]:=
```
AA = {{1, 0, 0, 0}, {0, -sigma, 0, 0}, {0, 0, 1/2 (sigma D3 + Sqrt[4 + D3^2]), 0}, {0, 0, 0, 1/2 (-D3 - sigma Sqrt[4 + D3^2])}};
```

```
BB = {{1, 0, 0, 0}, {0, -sigma, 0, 0}, {0, 0, 1/2 (sigma D3 - Sqrt[4 + D3^2]), 0}, {0, 0, 0, 1/2 (-D3 + sigma Sqrt[4 + D3^2])}};
```

In[44]:=
```
delta = frSym - (vars.AA.vars) (vars.BB.vars);
```

In[45]:=
```
exp = FullSimplify[Expand[delta //. dSub]]
```

Out[45]=
```
1/4 (-1 + sigma^2) (-(-2 x1^2 + D3 x2^2)^2 + (4 + D3^2) x3^4)
```

In[46]:=
```
exp /. sigma → 1
exp /. sigma → -1
```

Out[46]=
```
0
```

Out[47]=
```
0
```

■ Check signatures

In[48]:=
```
AA /. sigma → -1 // MatrixForm
AA /. sigma → 1 // MatrixForm
```

Out[48]//MatrixForm=

$$\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & \frac{1}{2}\left(-D3+\sqrt{4+D3^2}\right) & 0 \\ 0 & 0 & 0 & \frac{1}{2}\left(-D3+\sqrt{4+D3^2}\right) \end{pmatrix}$$

Out[49]//MatrixForm=

$$\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & -1 & 0 & 0 \\ 0 & 0 & \frac{1}{2}\left(D3+\sqrt{4+D3^2}\right) & 0 \\ 0 & 0 & 0 & \frac{1}{2}\left(-D3-\sqrt{4+D3^2}\right) \end{pmatrix}$$

In[50]:=
```
printNotebook["Metaclass_VI.pdf"]
```